# Maintainer: David Runge <dvzrv@archlinux.org>

pkgname=simde
pkgver=0.7.6
pkgrel=1
pkgdesc="Implementations of SIMD instruction sets"
arch=(any)
url="https://github.com/simd-everywhere/simde"
license=(MIT)
makedepends=(meson)
source=(
  $url/archive/v$pkgver/$pkgname-v$pkgver.tar.gz
)
sha512sums=('de17fca563c4db6766881e1c73142ad129a57febe55fa8ea1ae780226e60a84891b13d387e75574f2722d77e4013176e3c7dfaf17bccd8682b7d3d3ec8e92a54')
b2sums=('96378e81a2e2a2d797289d26b3d0c1ea72288b51094e8c5fde75ecbef34a0c1d4dc029d5974281e2f9f938b190b4a648225fcb486205cc496ba57b1ec528cee4')

build() {
  arch-meson build $pkgname-$pkgver
  meson compile -C build
}

check() {
  meson test -C build --print-errorlogs
}

package() {
  meson install -C build --destdir "$pkgdir"
  install -vDm 644 $pkgname-$pkgver/README.md -t "$pkgdir/usr/share/doc/$pkgname/"
  install -vDm 644 $pkgname-$pkgver/COPYING -t "$pkgdir/usr/share/licenses/$pkgname/"
}
